

Pianificazione Multiperiodo

Si vuole pianificare la produzione di tre prodotti A_1 , A_2 , A_3 , su un orizzonte temporale di quattro mesi, da Gennaio ad Aprile. La domanda cambia non solo da un prodotto a un altro, ma anche da un mese all'altro, come riportato nella tabella seguente.

Domanda di produzione	Gennaio	Febbraio	Marzo	Aprile
A_1	5300	1200	7400	5300
A_2	4500	5400	6500	7200
A_3	4400	6700	12500	13200

I prezzi di vendita, i costi di produzione e le quote di produzione sono come nella tabella seguente.

Produzione	A_1	A_2	A_3
Prezzo di vendita	\$124	\$109	\$115
Costo di produzione	\$73.30	\$52.90	\$65.40
Quota di produzione	500	450	550

I giorni di produzione disponibili variano ogni mese: 23 giorni a Gennaio, 20 a Febbraio, 23 a Marzo e 22 ad Aprile. Inoltre abbiamo la disponibilità di un magazzino con costo unitario mensile di \$3.50 per A_1 , \$4.00 per A_2 e \$3.00 per A_3 . Ogni prodotto occupa lo stesso volume e il volume totale disponibile nel magazzino è di 800 unità.

1. Formulare un modello AMPL, con relativo file di dati, per determinare il piano di produzione che massimizza il guadagno totale.
2. (*Opzionale*) Come cambiano: formulazione, modello e dati di AMPL se per ogni prodotto abbiamo un costo di attivazione e un lotto minimo da produrre, come riportato in tabella?

Produzione	A_1	A_2	A_3
Costo di attivazione	\$150000	\$150000	\$100000
Lotto minimo	20	20	16

La validità della decisione sull'attivazione o meno di una linea di produzione dura un mese, e i lotti minimi sono da intendersi mensili.

Soluzione

FORMULAZIONE.

- *Indici*:
 - sia i un indice sull'insieme $\pi = \{A_1, A_2, A_3\}$;
 - sia j un indice sull'insieme $\mu = \{1, 2, 3, 4\}$.
- *Parametri*: per ogni $i \in \pi, j \in \mu$:
 - sia P_j il numero di giorni di produzione nel mese j ;
 - sia d_{ij} la domanda massima per il prodotto i nel mese j ;
 - sia v_i il prezzo di vendita per il prodotto i ;
 - sia c_i il costo di produzione del prodotto i ;
 - sia q_i la quota di produzione massima del prodotto i ;
 - sia m_i il costo di magazzino per il prodotto i ;
 - sia C la capacità del magazzino in numero di pezzi;
 - sia a_i il costo di attivazione della produzione i (solo per la domanda 2);
 - sia l_i il lotto minimo di prodotto i (solo per la domanda 2).
- *Variabili*:
 - sia x_{ij} la quantità di prodotto i fatta nel mese j ;
 - sia w_{ij} la quantità di prodotto i venduta nel mese j ;
 - per ogni $i \in \pi, j \in \mu \cup \{0\}$, sia z_{ij} la quantità di prodotto i stoccata in magazzino nel mese j ;
 - sia $y_{ij} = 1$ se la produzione i viene attivata nel mese j , e 0 altrimenti (solo per la domanda 2).

Tutte le variabili sono non negative. Le variabili y_{ij} sono binarie, per ogni i e j (solo per la domanda 2).

- *Funzione Obiettivo*: (termine in y solo per la domanda 2)

$$\max \sum_{i=1}^3 \left(v_i \sum_{j=1}^4 w_{ij} - c_i \sum_{j=1}^4 x_{ij} - m_i \sum_{j=1}^4 z_{ij} - a_i \sum_{j=1}^4 y_{ij} \right).$$

- *Vincoli*:
 1. (richiesta): per ogni i, j : $w_{ij} \leq d_{ij}$;
 2. (produzione): per ogni j : $\sum_{i=1}^3 \frac{x_{ij}}{q_i} \leq P_j$;

3. (bilancio): per ogni i, j : $z_{i,j-1} + x_{ij} = z_{ij} + w_{ij}$;
4. (capacità): per ogni j : $\sum_{i=1}^3 z_{ij} \leq C$;
5. (dicembre): per ogni i : $z_{i0} = 0$;
6. (attivazione): per ogni i, j : $x_{ij} \leq P_j q_i y_{ij}$ (solo per la domanda 2);
7. (lotto minimo): per ogni i, j : $x_{ij} \geq l_i y_{ij}$ (solo per la domanda 2).

File modello in AMPL

```
# multiperiodo.mod - file modello per AMPL - produzione multiperiodo

set PRODOTTI;

param Mesi;

set MESI := 1..Mesi;
set MESIO := MESI union {0};

param giorni{MESI} >= 0;
param domanda { PRODOTTI, MESI } >= 0;
param vendita { PRODOTTI } >= 0;
param costo { PRODOTTI } >= 0;
param quota { PRODOTTI } >= 0;
param attivazione { PRODOTTI } >= 0;
param lotto { PRODOTTI } >= 0;
param magazzino { PRODOTTI } >= 0;
param capacita >= 0;

var x { PRODOTTI, MESI } >= 0;
var w { PRODOTTI, MESI } >= 0;
var z { PRODOTTI, MESIO } >= 0;
var y { PRODOTTI, MESI } >= 0, binary;

maximize guadagno:
sum {i in PRODOTTI}
(vendita[i] * sum {j in MESI} w[i,j] -
costo[i] * sum {j in MESI} x[i,j] -
magazzino[i] * sum {j in MESI} z[i,j] -
attivazione[i] * sum {j in MESI} y[i,j]) ;

subject to richiesta {i in PRODOTTI, j in MESI}:
w[i,j] <= domanda[i,j];

subject to produzione {j in MESI}:
sum {i in PRODOTTI} (x[i,j] / quota[i]) <= giorni[j];
```

```
subject to bilancio {i in PRODOTTI, j in MESI}:
z[i,j-1] + x[i,j] = z[i,j] + w[i,j];

subject to capacitamag {j in MESI}:
sum {i in PRODOTTI} z[i,j] <= capacita;

subject to attivaz {i in PRODOTTI, j in MESI}:
x[i,j] <= giorni[j]*quota[i]*y[i,j];

subject to lottominimo {i in PRODOTTI, j in MESI}:
x[i,j] >= lotto[i]*y[i,j];
```

File dati in AMPL

```
# multiperiodo.dat - file di dati per AMPL - produzione multiperiodo

set PRODOTTI := A1 A2 A3 ;

param Mesi := 4 ;

param giorni :=
  1 23
  2 20
  3 23
  4 22 ;

param domanda: 1 2 3 4 :=
A1 5300 1200 7400 5300
A2 4500 5400 6500 7200
A3 4400 6700 12500 13200 ;

param : vendita costo quota attivazione lotto magazzino :=
  A1 124 73.30 500 150000 20 3.5
  A2 109 52.90 450 150000 20 4
  A3 115 65.40 550 100000 16 3 ;

param capacita := 800 ;

let {i in PRODOTTI} z[i,0] := 0;
fix {i in PRODOTTI} z[i,0];
```

Soluzione

Solo per la domanda 2.

CPLEX 7.1.0: optimal integer solution; objective 1581550

47 MIP simplex iterations
0 branch-and-bound nodes
guadagno = 1581550.0000

x :=

A1 1	6100.0000
A1 2	0.0000
A1 3	0.0000
A1 4	0.0000
A2 1	0.0000
A2 2	3518.1818
A2 3	0.0000
A2 4	0.0000
A3 1	4400.0000
A3 2	6700.0000
A3 3	12650.0000
A3 4	12100.0000

;

y :=

A1 1	1.0000
A1 2	0.0000
A1 3	0.0000
A1 4	0.0000
A2 1	0.0000
A2 2	1.0000
A2 3	0.0000
A2 4	0.0000
A3 1	1.0000
A3 2	1.0000
A3 3	1.0000
A3 4	1.0000

;